﻿
CREATE PROCEDURE [dbo].[GoodsVsContragentsDelete]
(
	@Original_Id int,
	@Original_MD datetime,
	@MW varchar(50) -- если = PhysicallyDelete то запись удаляется физически. В противном случае поле ToDelete устанавливается в 1
)
AS
DECLARE @strMsg nvarchar(500), @intRet int, @intErrCode int
DECLARE @orgGoodsCode int
DECLARE @datMD datetime, @intDepartmentId int, @bitPhysicallyDelete bit

	SET NOCOUNT ON;

SELECT @orgGoodsCode = GoodsCode
FROM [CA].[GoodsVsContragents] WHERE Id = @Original_Id AND MD = @Original_MD
IF (@@rowcount <> 1)
BEGIN
	SELECT @intErrCode = 1, 
	@strMsg = 'Данные изменены после последнего запроса!'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF EXISTS (SELECT * FROM CD.InputLines WHERE SupplierGoodsId = @Original_Id)
BEGIN
	SELECT @intErrCode = 2,
	@strMsg = 'Существует приход для перекодировки с идентификатором ' + cast(@Original_Id as varchar(20)) + ' !'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

IF @Original_Id = 1
BEGIN
	SELECT @intErrCode = 3,
	@strMsg = 'Нельзя удалять перекодировку с идентификатором равным 1 !'
	RAISERROR (@strMsg, 11, 1)
	RETURN @intErrCode
END

--Проверка возможности физического удаления записи из таблиц справочников
EXEC @intRet = dbo.CheckDictionaryDelete @bitPhysicallyDelete OUTPUT, @intDepartmentId OUTPUT, @MW OUTPUT
IF @intRet <> 0 RETURN @intRet

EXEC [dbo].[ProcessStatistic] Null, @datMD OUTPUT, @MW OUTPUT

--удаляем товара из прайслиста поставщика
DELETE FROM PR.PriceLinesParsed WHERE GoodsVsContragentsId = @Original_Id

	SET NOCOUNT OFF;
IF @intDepartmentId <> 0 --в поздразделениях запись всегда физически удаляется
	DELETE FROM [CA].[GoodsVsContragents] WHERE Id = @Original_Id
ELSE
BEGIN
	IF @bitPhysicallyDelete = 1
		DELETE FROM [CA].[GoodsVsContragents] WHERE Id = @Original_Id
	ELSE --Обычные пользователи только помечают запись для удаления
		UPDATE [CA].[GoodsVsContragents] SET 
		ToDelete = 1,
		[MD] = @datMD, [MW] = @MW
		WHERE Id = @Original_Id
END

SET @intErrCode = @@error
IF @intErrCode <> 0
	IF @@trancount > 0 ROLLBACK TRANSACTION

RETURN @intErrCode
